

SAMPLE SOLUTION FOR:

Assignment: Linear (Integer) Programming – Improving Ambulance Service

620-461 Modelling of Business, Management and Industrial Problems

Semester 1, 2005

Prepared by
Associate Professor Natasha Boland & Dr. Heng-Soon Gan

NOTE: As it was not explicitly stated in the question whether the ARTG penalties are paid annually or as a one-off cost, both versions are accepted for assessment purposes. We will discuss both versions here.

(a) **CALCULATE** the current average response time in each district and the total ARTG penalty payable.

The current response time for each ASC, average response time and ARTG penalty are evaluated and tabulated in Table 1 and Table 2. The total ARTG penalty payable is **\$450,083** (rounded to the nearest dollar).

Table 1: Current response time for each ASC.

ASC	alpha(j)	n(j)	Response Time (minutes)
UC-1	2	3	12.5
UC-2	2	4	10
UC-3	2	3	12.5
UC-4	2	4	10
UC-5	2	3	12.5
UC-6	2	5	7.5
UC-7	1	2	5
UC-8	1	4	5
UC-9	2	5	7.5
UC-10	2	2	15

Table 2: Average response time and ARTG penalty incurred at each district.

		Response Time (minute)												
		UC-1	UC-2	UC-3	UC-4	UC-5	UC-6	UC-7	UC-8	UC-9	UC-10			
	Penalty Rate (per minute)	Time Guarantee (minute)	12.5	10	12.5	10	12.5	7.5	5	5	7.5	15	Average Response Time	ARTG Penalty
Atlantis-1	\$15,000.00	7	1	1									11.25	\$63,750.00
Atlantis-2	\$12,000.00	5		1	1								11.25	\$75,000.00
Atlantis-3	\$13,000.00	6				1		1					8.75	\$35,750.00
Atlantis-4	\$18,000.00	8	1									1	13.75	\$103,500.00
Atlantis-5	\$9,000.00	5				1	1				1		10.00	\$45,000.00
Atlantis-6	\$13,000.00	5					1			1			8.75	\$48,750.00
Atlantis-7	\$14,000.00	7						1			1	1	9.17	\$30,333.33
Atlantis-8	\$12,000.00	6			1			1					10.00	\$48,000.00
												TOTAL ARTG PENALTY	\$450,083.33	

(b) **FORMULATE**, in a general sense, a linear integer program whose solution will tell UAS how to minimise the total cost. Define clearly all parameters and variables used.

- Let d be the number of districts
- m be the number of ASC's
- S_i be the set of ASC within 10km radius from the center district i .
- n be the maximum number of upgrades for any district
- a_j be the number of ambulances currently available at ASC j
- \bar{C}_j be the capacity of the upgraded ASC j
- c be the cost of purchasing an ambulance
- f_j be the cost of upgrading ASC j
- T_i be the ARTG at district i
- p_i be the penalty per minute exceeding ARTG, T_i at district i
- B be ASC's budget
- \bar{p} be the penalty per dollar for exceeding B

Decision variables: $y_j = \begin{cases} 1, & \text{if ASC } j \text{ is upgraded} \\ 0, & \text{otherwise} \end{cases}$

- x_j is the number of new ambulances purchased at ASC j
- r_i^+ is the average response time exceeding ARTG in district i
- R_j be the response time of ASC j
- b^+ is the amount exceeding budget B

LIP MODEL:

$$\text{Minimise } \bar{p}b^+ + \sum_{i=1}^d p_i r_i^+$$

s.t.

$$\sum_{j \in S_i} y_j \leq n, \quad i = 1 \dots d$$

$$R_j \geq 5, \quad j = 1 \dots m$$

$$R_j \geq \left[10\alpha_j - \frac{5}{\alpha_j}(x_j + a_j) \right], \quad j = 1 \dots m$$

$$\frac{\sum_{j \in S_i} R_j}{|S_i|} - T_i \leq r_i^+, \quad i = 1 \dots d$$

$$x_j \leq y_j(\bar{C}_j - a_j), \quad j = 1 \dots m$$

$$\sum_{j=1}^m (cx_j + f_j y_j) - B \leq b^+$$

$$y_j \in \{0,1\}, \quad j = 1 \dots m$$

$$x_j \text{ integer}, \quad j = 1 \dots m$$

$$r_i^+ \geq 0, \quad i = 1 \dots d$$

$$b^+ \geq 0$$

(c) **CODE** and **SOLVE** the formulation in part (b) using Xpress^{MP} using the sample code at <http://www.more.ms.unimelb.edu.au/toolsandtechniques/educationsamples/educationsamples.htm>. **COMMENT** on the result.

The Mosel code and datafile are shown below:

```
model "Ambulance"
  uses "mmsxprs"

  parameters
    DATAFILE = "ambulance.dat"
  end-parameters

  declarations
    NUM_DISTRICT: set of string
    NUM_CENTRE: set of string

    Max_Upgrade: real
    Ambulance_Cost: real
    Budget: real
    Budget_Penalty: real
    Response_Rate: real
    Response_Intercept: real
    Min_Response: real

    District_Centre: dynamic array(NUM_DISTRICT,NUM_CENTRE) of integer
    Current_Num_Ambulance: array(NUM_CENTRE) of real
    New_Capacity: array(NUM_CENTRE) of real
    Upgrade_Fixed_Cost: array(NUM_CENTRE) of real
    ARTG: array(NUM_DISTRICT) of real
    ARTG_Penalty: array(NUM_DISTRICT) of real
    Alpha: array(NUM_CENTRE) of real
  end-declarations

  initializations from DATAFILE
    Max_Upgrade
    Ambulance_Cost
    Budget
    Budget_Penalty
    Response_Rate
    Response_Intercept
    Min_Response

    District_Centre
    Current_Num_Ambulance
    New_Capacity
    Upgrade_Fixed_Cost
    ARTG
    ARTG_Penalty
  end-initializations

  finalize(NUM_DISTRICT)
  finalize(NUM_CENTRE)

  declarations
    Upgrade_Or_Not: array(NUM_CENTRE) of mpvar
    Num_New_Ambulance: array(NUM_CENTRE) of mpvar
    Under_Time: array(NUM_DISTRICT) of mpvar
    Over_Time: array(NUM_DISTRICT) of mpvar
    Under_Budget: mpvar
    Over_Budget: mpvar
    Response_Time: array(NUM_CENTRE) of mpvar
  end-declarations
```

```

forward procedure print_solution

forall(j in NUM_CENTRE) do
  Alpha(j) := sum(i in NUM_DISTRICT | exists(District_Centre(i,j)))
             District_Centre(i,j)
end-do

Total_Penalty := Budget_Penalty * Over_Budget +
  sum(i in NUM_DISTRICT) ARTG_Penalty(i) * Over_Time(i)

forall(i in NUM_DISTRICT) sum(j in NUM_CENTRE | exists(District_Centre(i,j)))
  Upgrade_Or_Not(j) <= Max_Upgrade

forall(j in NUM_CENTRE) do
  Response_Time(j) >= Response_Intercept*Alpha(j) -
    Response_Rate*(Current_Num_Ambulance(j) + Num_New_Ambulance(j))/Alpha(j)

  Response_Time(j) >= Min_Response
end-do

forall(i in NUM_DISTRICT) do
  (sum(j in NUM_CENTRE | exists(District_Centre(i,j)))
   Response_Time(j))/sum(j in NUM_CENTRE | exists(District_Centre(i,j)))
   District_Centre(i,j) - ARTG(i) <= Over_Time(i)
end-do

forall(j in NUM_CENTRE) Num_New_Ambulance(j) <=
  Upgrade_Or_Not(j) * (New_Capacity(j) - Current_Num_Ambulance(j))

sum(j in NUM_CENTRE) (Ambulance_Cost * Num_New_Ambulance(j) +
  Upgrade_Fixed_Cost(j) * Upgrade_Or_Not(j)) - Budget <= Over_Budget

forall(j in NUM_CENTRE) Upgrade_Or_Not(j) is_binary
forall(j in NUM_CENTRE) Num_New_Ambulance(j) is_integer

minimise(Total_Penalty)

print_solution

procedure print_solution
  writeln("Total penalty: $", getobjval)

  writeln("Amount exceeding budget: $", getsol(Over_Budget))

  writeln("ARTG Penalty: $", getsol(sum(i in NUM_DISTRICT)
    ARTG_Penalty(i) * Over_Time(i)))

  writeln("Upgrade cost: $", getsol(sum(j in NUM_CENTRE)
    (Ambulance_Cost * Num_New_Ambulance(j) +
    Upgrade_Fixed_Cost(j) * Upgrade_Or_Not(j))))

  writeln

```

```

writeln("Centre ", " Upgrade? ", " Response Time ",
" Alpha ", " Current Num ", " Ambulance Added ")
forall(j in NUM_CENTRE) do
  write(strfmt(j, 6))
  write(strfmt(getsol(Upgrade_Or_Not(j)), 8))
  write(strfmt(getsol(Response_Time(j)), 13))
  write(strfmt(getsol(Alpha(j)), 8))
  write(strfmt(Current_Num_Ambulance(j), 14))
  write(strfmt(getsol(Num_New_Ambulance(j)), 14))
  writeln
end-do
writeln

writeln("District ", " Ave Response ", " Target ")
forall(i in NUM_DISTRICT) do
  write(strfmt(i, 8))
  write(strfmt(getsol((sum(j in NUM_CENTRE | exists(District_Centre(i,j)))
  Response_Time(j)) / (sum(j in NUM_CENTRE | exists(District_Centre(i,j)))
  District_Centre(i,j))), 11))
  write(strfmt(ARTG(i), 8))
  writeln
end-do
end-procedure

end-model

```

!Datafile for "ambulance.mos"

Max_Upgrade: 1

Ambulance_Cost: 70000

Budget: 1000000

Budget_Penalty: 1

Response_Rate: 5

Response_Intercept: 10

Min_Response: 5

Current_Num_Ambulance: [

("UC-1") 3
 ("UC-2") 4
 ("UC-3") 3
 ("UC-4") 4
 ("UC-5") 3
 ("UC-6") 5
 ("UC-7") 2
 ("UC-8") 4
 ("UC-9") 5
 ("UC-10") 2

]

New_Capacity: [

("UC-1") 10
 ("UC-2") 20
 ("UC-3") 18
 ("UC-4") 17
 ("UC-5") 12

```
    ("UC-6") 13
    ("UC-7") 20
    ("UC-8") 18
    ("UC-9") 20
    ("UC-10") 11
]

Upgrade_Fixed_Cost: [
    ("UC-1") 100000
    ("UC-2") 120000
    ("UC-3") 160000
    ("UC-4") 180000
    ("UC-5") 130000
    ("UC-6") 100000
    ("UC-7") 180000
    ("UC-8") 150000
    ("UC-9") 180000
    ("UC-10") 120000
]

ARTG: [
    ("Atlantis-1") 7
    ("Atlantis-2") 5
    ("Atlantis-3") 6
    ("Atlantis-4") 8
    ("Atlantis-5") 5
    ("Atlantis-6") 5
    ("Atlantis-7") 7
    ("Atlantis-8") 6
]

ARTG_Penalty: [
    ("Atlantis-1") 15000
    ("Atlantis-2") 12000
    ("Atlantis-3") 13000
    ("Atlantis-4") 18000
    ("Atlantis-5") 9000
    ("Atlantis-6") 13000
    ("Atlantis-7") 14000
    ("Atlantis-8") 12000
]

District_Centre: [
    ("Atlantis-1" "UC-1") 1
    ("Atlantis-1" "UC-2") 1
    ("Atlantis-2" "UC-2") 1
    ("Atlantis-2" "UC-3") 1
    ("Atlantis-3" "UC-4") 1
    ("Atlantis-3" "UC-6") 1
    ("Atlantis-4" "UC-1") 1
    ("Atlantis-4" "UC-10") 1
    ("Atlantis-5" "UC-4") 1
    ("Atlantis-5" "UC-5") 1
    ("Atlantis-5" "UC-9") 1
    ("Atlantis-6" "UC-5") 1
    ("Atlantis-6" "UC-8") 1
    ("Atlantis-7" "UC-7") 1
    ("Atlantis-7" "UC-9") 1
    ("Atlantis-7" "UC-10") 1
    ("Atlantis-8" "UC-3") 1
    ("Atlantis-8" "UC-6") 1
]
```

Xpress-IVE output:

Total penalty: \$185083
Amount exceeding budget: \$20000
ARTG Penalty: \$165083
Upgrade cost: \$1.02e+006

Centre	Upgrade?	Response Time	Alpha	Current Num	Ambulance Added
UC-1	1	5	2	3	3
UC-2	0	10	2	4	0
UC-3	1	5	2	3	3
UC-4	0	10	2	4	0
UC-5	1	5	2	3	3
UC-6	0	7.5	2	5	0
UC-7	0	5	1	2	0
UC-8	0	5	1	4	0
UC-9	0	7.5	2	5	0
UC-10	0	15	2	2	0

District	Ave Response	Target
Atlantis-1	7.5	7
Atlantis-2	7.5	5
Atlantis-3	8.75	6
Atlantis-4	10	8
Atlantis-5	7.5	5
Atlantis-6	5	5
Atlantis-7	9.16667	7
Atlantis-8	6.25	6

Ambulance service centres **UC-1, UC-3 and UC-5** should be upgraded with **three new ambulances** purchased for each. A total of **\$1,020,000** will be incurred for the upgrade. The ARTG penalty payable is **\$165,083** and the penalty for exceeding the budget is **\$20,000**.

(d) *Should UAS proceed with the upgrade as suggested in part (c) and sign the agreement, if it was a 10-year quality-of-service agreement and no upgrades are allowed for the duration of the agreement?*

We will ignore the effect of time value of money in the following analysis and assume that the ARTG penalty rates do not change for the duration of the contract, i.e. for 10 years. For the purpose of this analysis, we will also ignore the operating costs of the ambulances (NOTE: The ambulance operating costs can be embedded into the LIP model. Realistically, you may wish to distinguish operating costs of old and new ambulances.)

Suppose that the ARTG penalty is payable annually, for the 10-year period:

If upgrades are not carried out, the total expenditure at the end of the 10-year period is
= 10 x \$450,083
= \$4,500,830

If upgrades were to be carried out, the total expenditure at the end of the 10-year period is
 = \$1,020,000 + \$20,000 + 10 x \$165,083
 = \$2,690,830

From the calculations above, UAS should proceed with the upgrade. By doing so, UAS is able to save \$1,810,000 of expenditure over a 10-year period, despite initial expenditure of \$1,020,000 for the upgrades and budget penalty of \$20,000.

Now suppose that the ARTG penalty is payable once, and if upgrades were to be carried out, the total expenditure is

= \$1,020,000 + \$20,000 + \$165,083
 = \$1,205,083 (> \$450,083 if no upgrade is carried out)

Hence, for the case where the ARTG penalty is only payable once, we recommend that UAS do not proceed with the upgrade.

(e) UAS wish to evaluate the effect of the political decision to restrict at most one ASC in any district to be upgraded. **REFORMULATE** and **SOLVE** the linear integer program when at most two ASC's in each district can be upgraded. **ILLUSTRATE** and **COMPARE** both results in parts (c) and (e). Do the same for the case for any number of ASC's in a district could be upgraded. Does anything change?

No reformulation is needed here. We only need to let $n = 2$ for the where at most two ASC's in each district can be upgraded. Solving the LIP with Xpress^{MP} yields:

Total penalty: \$177000
 Amount exceeding budget: \$0
 ARTG Penalty: \$177000
 Upgrade cost: \$940000

Centre	Upgrade?	Response Time	Alpha	Current Num	Ambulance Added
UC-1	1	5	2	3	3
UC-2	0	10	2	4	0
UC-3	1	5	2	3	3
UC-4	0	10	2	4	0
UC-5	0	12.5	2	3	0
UC-6	0	7.5	2	5	0
UC-7	0	5	1	2	0
UC-8	0	5	1	4	0
UC-9	0	7.5	2	5	0
UC-10	1	10	2	2	2

District	Ave Response	Target
Atlantis-1	7.5	7
Atlantis-2	7.5	5
Atlantis-3	8.75	6
Atlantis-4	7.5	8
Atlantis-5	10	5
Atlantis-6	8.75	5
Atlantis-7	7.5	7
Atlantis-8	6.25	6

When at most two ASC's in each district can be upgraded, UC-1, UC-3 and UC-10 are upgraded, with 3 new ambulances at UC-1 and UC-3 each and 2 new ambulances at UC-10. A total of **\$940,000** will be incurred for the upgrade, the ARTG penalty payable is **\$177,000** and no penalties for exceeding the budget.

As a comparison to the case where at most one ASC can be upgraded (part (c)), this is a savings of \$100,000 on the upgrade cost and budget penalty, but an increment by \$11,917 on the ARTG penalty. The upgrade cost and budget penalty savings and ARTG penalty increment imply that relaxing the maximum number of upgrades in each district is favourable if the ARTG penalty is only paid once (see Table 3). However, we do not see any financial benefit from the relaxation if the ARTG penalty is paid annually. But these arguments are based on the fact that we have ignored operating costs of the ambulances. The ambulance operating costs for the case when $n = 2$ may be lower since a total of 7 new ambulances are only needed, as compared to a total of 9 ambulances for the case when $n = 1$. The lower ambulance operating costs for $n = 2$ may then justify the relaxation.

Increasing the maximum number of upgrades beyond two does not alter the solution at all. A summary of the results is shown in Table 3 and graphical representations of the solutions are shown in Figure 1 and Figure 2.

Table 3: Summary of results for different n values.

	Maximum number of upgrades in each district			
	0 (no upgrades)	1	2	3
ARTG Penalty	\$450,083	\$165,000	\$177,000	\$177,000
Total Upgrade Cost	\$0	\$1,020,000	\$940,000	\$940,000
Budget Penalty	\$0	\$20,000	\$0	\$0
Upgrade Decision [UC-#(number of new ambulances)]	-	UC-1 (3) UC-2 (3) UC-5 (3)	UC-1 (3) UC-3 (2) UC-10 (2)	UC-1 (3) UC-3 (2) UC-10 (2)
Total expenditure Case: 10-year contract, ARTG penalty paid annually for 10 years	\$4,500,830	\$2,690,830	\$2,710,000	\$2,710,000
Total expenditure Case: 10-year contract, ARTG penalty paid once	\$450,083	\$1,205,083	\$1,117,000	\$1,117,000




Figure 1: Graphical representation of the solution when $n = 1$. The shaded boxes are ASCs to be upgraded.




Figure 2: Graphical representation of the solution when $n = 2$. The shaded boxes are ASCs to be upgraded.